
Ein Werkzeug für das Reporting – ohne Makros 

Ein Beispiel 

1. Der Hintergrund 

Bei der Landeshauptstadt München wurde bislang libreOffice Calc eingesetzt. Dem 
Kommunalreferat steht eine Datenbank zur Verfügung, die eine Liste der Immobilien exportiert, 
aufbereitet und „per Hand“ weiter verarbeitet wird: 

 

Die Filterung wird per Hand vorgenommen, das Filterkriterium in den Kopf des Tabellenblattes 
eingetragen. Mögliche nicht passende Daten werden per Hand aus der Liste entnommen. Diese 
Ergebnisse werden in ein weiteres Blatt kopiert und dort für einen Bericht aufgearbeitet: 

2. Die Aufgabe 

Bei der Umstellung auf Microsoft Excel überlegten sich die Verantwortlichen das Werkzeug mit 
den Werkzeugen von Excel aufzubereiten. Und hier meine Lösung: 

3. Die Daten 

In regelmäßigen Abständen erfolgt ein Datendownload. Diese Excelmappe wird in einem Ordner 
gespeichert. Greift man mit PowerQuery darauf zu, steht der Pfad „hart“ codiert im Code: 

 

 

Der Dateiname und der Pfad werden in eine Zelle ausgelagert, in Excel in eine (intelligente) 
Tabelle verwandelt: 

 

Die Tabellen heißen tbl_Pfad und tbl_Dateiname. Diese werden nach PowerQuery als Tabelle 
„gezogen“ und dort wird die letzte Zeile per Drilldown als „Variable“, das heißt als Text 
gespeichert: 

 

So können diese Variablen im M-Code verwendet werden. Und so wird der Datenzugriff 
dynamisch: 

Diese Tabelle wird qry_Datenzugriff genannt und nach Excel zurückgegeben. 

Hinweis 

Dadurch, dass nun der Code zusammengebaut wird, greifen die Sicherheitseinstellungen 
von PowerQuery. Man erhält eine Fehlerzeile: 

 

Diese Optionen können in Daten / Daten abrufen / Abfrageoptionen können deaktiviert 
werden – dort in der aktuellen Arbeitsmappe muss man beim Datenschutz die Sicherheitsstufen 
ignorieren: 

 

4. Filter: Spalten und Zeilen 

Damit der Zugriff dynamisch bleibt werden zwei Filter eingefügt: einmal auf die Spalten, einmal 
auf die Zeilen. 

Alle Spaltennamen werden aufgelistet. Hierzu leistet die Funktion BEREICH.VERSCHIEBEN 
gute Dienste: 

=BEREICH.VERSCHIEBEN(Daten!$A$9;1;ZEILE(A1)-1) 

 

Diese Tabelle wird nach PowerQuery „gezogen“, dort die ausgewählten Zeilen gefiltert und in 
eine Liste konvertiert (per Drilldown) 

Ebenso wird eine Ausschluss-Tabelle erzeugt: 

 

Beide Tabellen bilden die Grundlage für das Filtern (nach Zeile und nach Spalte) 

Auf die erste Tabelle (qry_Datenzugriff) wird eine Verknüpfung hergestellt. 

Die „Spaltentabelle“ (tbl_Kriterien) wird im Code verwendet: 

= Table.SelectColumns(Quelle,tbl_Kriterien) 

Die zweite Tabellen wird für eine Verknüpfung verwendet: 

 

Wichtig: der linke Anti-Join sorgt dafür, dass diese Zeilen aus der ursprünglichen Liste gefiltert 
werden. 

Die letzte Spalte kann gelöscht werden: 

Das Ergebnis wird nach Excel zurück geschrieben und bildet die Datenbasis: 

 

Hinweis 

Da in der Spalte Registrierungsnummer die ID, also der Schlüsselwert steht, darf dieser 
nicht gelöscht werden. 

 

5. Weitere Filter 

Ich benötige weitere Filter.  

Excel stellt folgende Filter zur Verfügung: 

 Der Autofilter. Er zeigt beim Darüberfahren das Filterkriterium an. 
 Der Spezialfilter („erweitert“). Er muss beim Ändern der Kriterien erneut aufgerufen 

werden. 
 Der Datenschnitt. Er setzt eine (intelligente) Tabelle voraus und kann nicht 

Zahlenbereiche (> oder <) filtern. 
 Filterungen über PowerQuery. Man muss die Listen aktualisieren! 
 Die Funktion FILTER 

Die Funktion FILTER holt die entsprechenden Daten aus der Liste. Sie wird in der Form  

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
qry_Datenzugriff_SpaltenAuswahl[Lagekarte-GAA]="B - Beste") 

geschrieben. Lagert man den Text B – Beste in die Zelle B2 aus, kann man die Funktion 
schreiben: 

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
qry_Datenzugriff_SpaltenAuswahl[Lagekarte-GAA]=B2) 

Lagert man den Text Lagekarte-GAA in die Zelle A2 aus, muss man verketten: 

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&A2&"]")=B2) 

Will man zwei Kriterien mit einem logischen UND verketten muss man schreiben: 

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
(qry_Datenzugriff_SpaltenAuswahl[Lagekarte-GAA]="B - Beste")* 
(qry_Datenzugriff_SpaltenAuswahl[Baujahr Geb.]<=1923)) 

Oder ausgelagert: 

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&A2&"]")=B2)* 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&D2&"]")<=F2)) 

Beim Verknüpfungsoperator ZWISCHEN muss man natürlich schreiben: 

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&A2&"]")=B2)* 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&D2&"]")>=E2)) 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&D2&"]")<=F2)) 

Sollte nun eine der Zellen leer sein, soll der Filter keine Anwendung finden. Da WAHR dem 
Wert 1 entspricht, multipliziert man die vorhandenen Werte mit 1, also: 

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
(WENN(ODER(A2="";B2=""); 
SEQUENZ(ZEILE(qry_Datenzugriff_SpaltenAuswahl);1;1;0); 
qry_Datenzugriff_SpaltenAuswahl[Lagekarte-GAA]=B2))* 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&D2&"]")<=F2)) 

Und für beide Kriterien: 

=FILTER(qry_Datenzugriff_SpaltenAuswahl; 
(WENN(ODER(A2="";B2=""); 
SEQUENZ(ZEILE(qry_Datenzugriff_SpaltenAuswahl);1;1;0); 
qry_Datenzugriff_SpaltenAuswahl[Lagekarte-GAA]=B2))* 
WENN(ODER(D2="";E2="";F2=""); 
SEQUENZ(ZEILE(qry_Datenzugriff_SpaltenAuswahl);1;1;0); 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&D2&"]")<=F2)* 
(INDIREKT("qry_Datenzugriff_SpaltenAuswahl["&D2&"]")>=E2))) 

Und das Ganze kann man für mehrere Kriterien ausbauen: 

Leider liefert die Funktion FILTER nicht die Überschrift. Sie kann in einer zweiten Funktion 
ermittelt werden: 

=qry_Datenzugriff_SpaltenAuswahl[#Kopfzeilen] 

Nun kann man Anzahl, Mittelwert, Median, … berechnen: 

=TEILERGEBNIS(2;BEREICH.VERSCHIEBEN(D$10;1;0; 
ZEILEN(qry_Datenzugriff_SpaltenAuswahl);1)) 

Oder auch direkt mit Anzahl, da die Funktion FILTER ja die Daten gefiltert hat: 

=ANZAHL(BEREICH.VERSCHIEBEN(D10;1;0;ZEILEN(A11#);1)) 

Hinweis 

Leider kann mit dieser Berechnung keine ODER-Filterung durchgeführt werden. 

Hinweis 

Eine Sortierung ist nicht möglich. Man muss die Originalliste sortieren. 

 

6. Die „Köpfe“ 

Im Kopfbereich befinden sich die Berechnungen, beispielsweise 

=TEILERGEBNIS(2;BEREICH.VERSCHIEBEN(F$17;1;0; 
ZEILEN(qry_Datenzugriff_SpaltenAuswahl);1)) 

7. Die Reports 

In einer weiteren Datei werden auf Basis dieser Daten die Reports erzeugt.  

Dort werden die „Köpfe“ und die gefilterten Daten in einer intelligenten Tabelle über 
PowerQuery zusammengefasst: 

 

Die Berichte verwenden diese Daten. XVERWEIS leistet gute Dinge: 

 

Auch ein automatischer Umbruch kann mit Formeln und bedingter Formatierung erzeugt 
werden: 

